Riešenie nelineárnych rovníc $f(x) = 0$

Príklad 1

- **Zadanie príkladu: Nájdite riešenie rovnice $x^2 + \ln x = \frac{10}{x}$ na intervale [0.1, 4].**

Zadanie úlohy typu "nájdite riešenie nelineárnej rovnice $x^2 + \ln x = \frac{10}{x}$ " sa skladá z nasledovných čiastkových úloh. Aj keď nie sú jednotlivé body striktne definované, napr. separácia, sú potrebné pre správne riešenie úlohy a preto ich musíme realizovať.

Urobte separáciu koreňov rovnice $x^2 + \ln x = \frac{10}{x}$ na intervale [0.1, 4], určite ich počet a násobnosť.

Metóda tetív.
S toleranciou 10^{-10} nájdite približnú hodnotu najväčšieho koreňa metódou lineárnej interpolácie - metóda tetív.
Vypočítajte odhad absolútnej chyby aproximácie.
Výsledok porovnajte s hodnotou koreňa nájdeného pomocou príkazu FindRoot [ ] (použite parametre príkazu pre metódu tetív)
NEZABUDNITE overiť splnenie podmienok použiteľnosti danej metódy, uveďte štartovacie body, uveďte iteračnú schému, prípadne iné obmedzujúce podmienky, ktoré boli použité v rámci výpočtu.
Uveďte dvojicu bodov, ktorú nie je vhodné použiť ako štartovacie body pre metódu tetív. Zdôvodnite prečo táto dvojica bodov nie je vhodná.

Newtonova metóda
S toleranciou 10^{-10} nájdite približnú hodnotu najväčšieho koreňa Newtonovou metódou.
Vypočítajte odhad absolútnej chyby aproximácie.
Výsledok porovnajte s hodnotou koreňa nájdeného pomocou príkazu FindRoot [ ] (použite parametre príkazu pre metódu tetív)
NEZABUDNITE overiť splnenie podmienok použiteľnosti danej metódy, uveďte štartovacie body, uveďte iteračnú schému, prípadne iné obmedzujúce podmienky, ktoré boli použité v rámci výpočtu.
Uveďte aspoň jeden bod, ktorý nie je vhodné použiť ako štartovací bod pre Newtonovu metódu. Zdôvodnite prečo táto voľba štartovacieho bodu nie je vhodná.

Na záver riešenia urobte analýzu problému a vysvetlite, ktoré metódy je/nie je vhodné použiť na riešenie tohto problému a prečo. Porovnajte obe metódy. Ktorá z nich rýchlejšie konverguje? Závery robte na základe konkrétnych výsledkov predchádzajúcich výpočtov (uveďte ktoré výsledky ste použili).

- **Riešenie:**

Rovnicu upravíme do tvaru $f(x) = 0$.

$$x^2 + \ln x = \frac{10}{x}$$

$$x^2 + \ln x - \frac{10}{x} = 0$$

V ďalšom výpočte budeme pracovať s funkciou $f(x) = x^2 + \ln x - \frac{10}{x}$. Funkciu $f(x)$ si nakreslíme.

```
In[1]:= f[x_] = x^2 + Log[x] - 10 / x
```

Out[1]= $-\frac{10}{x} + x^2 + \text{Log}[x]$

Pozrieme sa na celkový charakter funkcie $f(x)$.

In[2]:= `Plot[f[x], {x, 0.1, 10}]`

Out[2]=


Urobíme separáciu koreňov rovnice. Funkciu $f(x)$ nakreslíme na intervale [0.1, 4].

In[3]:= `Plot[f[x], {x, 0.1, 4}]`

Out[3]=


Z grafu funkcie vidíme, že funkcia $f(x)$ na intervale [0.1, 4] je spojitá a má práve jeden koreň. Interval môžeme zúžiť. Na základe obrázku vidíme, že funkcia $f(x)$ má práve jeden koreň na intervale [1.5, 2.5].

Ak by sme potrebovali formálne overiť, že funckai skutočne mení znamienko na intervale [1.5, 2.5], môžeme použiť jednoduchú podmienku: ak $f(a) * f(b) < 0$, potom funkcia $f(x)$ na intervale $[a, b]$ aspoň raz zmení znamienko. V našom prípade stačí overiť, či $f(1.5) * f(2.5) < 0$.

```
f[1.5] * f[2.5] < 0
```

Out[5]= `True`

■ Metóda tetív

Na základe separácie koreňov vieme, že na intervale [1.5, 2.5] sa nachádza práve jeden koreň.
Najskôr je potrebné overiť splnenie podmienok riešiteľnosti na tomto intervale

- $f(x)$ je spojitá na [1.5, 2.5].
- $f(x)$ je monotónna na [1.5, 2.5], buď rastúca alebo klesajúca na celom intervale, kde úlohu riešime. V našom prípade je $f(x)$ je rastúca.
- na intervale, kde úlohu riešime sa nachádza PRÁVE JEDEN KOREŇ.

Vidíme to z grafu funkcie, alebo to môžeme overiť splnením podmienky $f(1.5) * f(2.5) < 0$

Ak by sme nechceli dodržiavať formalizmus, inými slovami môžeme povedať: v okolí koreňa je funkcia $f(x)$ spojitá a nemá extrém ani inflexný bod.

Pri metóde tetív je potrebné zvoliť dva štartovacie body, iteračná schéma má tvar

$x_0 =$ ....

$x_1 =$ ....

$$x_{i+1} = x_i - f(x_i).\ \frac{x_i - x_{i-1}}{f(x_i) - f(x_{i-1})}$$

Podmienky použitia metódy budú možno porušené v okolí bodu 2.5 - vidíme to z grafu funkcie, ktorý sme nakreslili na intervale [0.1, 10], preto je lepšie voliť štartujúce body menšie ako 2. Pre ďalší výpočet zvolíme štartujúce hodnoty $x_0 = 1$, $x_1 = 1.5$.

In[6]:=
```
Plot[f[x], {x, 1.5, 2.5}]
```

Out[6]=



Vstupné údaje:

`f[x_]`	definovanie funkcie,
`x[0], x[1]`	štartovacie body
`pocetopakovani`	počet iterácií v cykle
`tolerancia`	presnosť, s ktorou chceme počítať

Úloha vytvor program:

Cyklus [iteračná schéma;

zastavovacia podmienka $| x[i+i] - x[i] | <$ tolerancia, zastav cyklus],

{iračný predpis pre riadiacu premennú cyklu}]

Po prebehnutí cyklu necháme vypočítané hodnoty vypísať do tabuľky.

In[39]:=
```
Clear[f, x]
f[x_] = x^2 + Log[x] - 10 / x;
```

In[41]:=
```
x[0] = 1;
x[1] = 1.5;
pocetopakovani = 10;
tolerancia = 10^(-10);
```

In[45]:=
```
Do[x[i + 1] = x[i] - f[x[i]] * (x[i] - x[i - 1]) / (f[x[i]] - f[x[i - 1]]) // N;
 If[Abs[x[i + 1] - x[i]] < tolerancia, Break[]],
 {i, 1, pocetopakovani}]
```

In[46]:=
```
TableForm[Table[{i, NumberForm[x[i], 10], Abs[x[i + 1] - x[i]]}, {i, 0, pocetopakovani}],
     TableHeadings → {None, {"i", "xi", "|xi+1-xi|"}},
     TableSpacing → {1, 5}]
```

Out[46]//TableForm=

i	x_i	$\lvert x_{i+1}-x_i \rvert$
0	1	0.5
1	1.5	0.402021
2	1.902020808	0.132969
3	2.034990198	0.00887101
4	2.043861213	0.0000703651
5	2.043931578	2.67468×10^{-8}
6	2.043931605	7.86038×10^{-14}
7	2.043931605	Abs[-2.04393 + x[8]]
8	x[8]	Abs[-x[8] + x[9]]
9	x[9]	Abs[-x[9] + x[10]]
10	x[10]	Abs[-x[10] + x[11]]

Kvôli istote sa pozrime ešte na presný tvar koreňa, pomocou príkazu: `NumberForm [číslo, počet miest]`

In[48]:=
```
NumberForm[x[6], 16]
```

Out[48]//NumberForm=
```
2.043931605061835
```

Výpočet realizovaný metódou tetív nám našiel približnú hodnotu koreňa rovnice $x_6 =$ 2.053931605061835. Uvedomte si, že je to len PRIBLIŽNÁ hodnota koreňa, skutočnú hodnotu koreňa nepoznáme, vieme len, že je veľmi blízko tejto vypočítanej hodnoty.
Vypočítajme preto odhad absolútnej chyby aproximácie podľa vzorca pre výpočet odhadu chyby. Tento vzorec nám dáva informáciu o tom, ako blízko sa nachádza nami vypočítaná približná hodnota koreňa x_6 od skutočného koreňa rovnice. Použijeme vzťah z prednášky.

$$|x_i - \alpha| \leq \frac{|f(x_i)|}{m}, \qquad \text{kde } m = \min \left| f'(x_i) \right| \text{ na intervale } [a, b]$$

Najskôr si nakreslíme deriváciu funkcie $f(x)$ na intervale, na ktorom realizujeme výpočet - [1.5, 2.5].

In[22]:=
```
Plot[f'[x], {x, 1.5, 2.5}]
```

Out[22]=


Vidíme, že na intervale [1.5, 2.5] derivácia funkcie $- \ f'(x)$ nadobúda najmenšiu hodnotu niekde v okolí bodu 2.217. Zistili sme to systémom zistím a vidím, odčítaním pomocou zamerania bodu z intervalu.

In[17]:=
```
f'[2.217] // N
```

Môžeme použiť aj funkciu programu *Mathematica* na nájdenie minima - to bude určite presnejšie, ako náš odhad. My ale hľadáme odhad, takže na presnosti až tak veľmi nezáleží a príkaz *Mathematice* môže v mnohých situáciách zlyhať.

In[23]:=
```
FindMinimum[f'[x], {x, 1.5, 2.5}]
```

Out[23]= {6.91933, {x → 2.23176}}

Vypočítajme odhad absolútnej chyby aproximácie podľa vzorca pre výpočet odhadu chyby - dosadíme do nasledovného vzťahu.

$$|x_i - \alpha| \leq \frac{|f'(x_i)|}{m}, \text{ kde } m = \min \left| f'(x_i) \right| \text{ na intervale } [a, b]$$

In[24]:=
```
Abs[f[x[6]]] / f'[2.217]
```

Out[24]= 7.91159×10^{-14}

Vidíme, že nami vypočítaná aproximácia koreňa rovnice sa od skutočného koreňa líši rádo 10^{-14}, čo je viac ako bola požadovaná presnosť výpočtu a preto môžeme považovať nami vypočítanú aproximáciu koreňa za použiteľnú.

Nami vypočítaný výsledok môžeme porovnať aj s riešením nájdeným pomocou príkazu FindRoot [ ]. Je to príkaz systému *Mathematica*. Ak parametre výpočtu je potrebné uviesť interval - koncové body, v ktorom sa nachádza práve jeden koreň.

In[26]:=
```
FindRoot[f[x], {x, 1.5, 2.5}]
```

Out[26]= {x → 2.04393}

V zadaní ešte žiadali nájsť dvojicu bodov, ktorá nie je vhodnou voľbou štartovacích bodov pre metódu tetív - 0.1 a 0.2. - extrém. Pri METÓDE TETÍV nie je vhodné voliť štartovacie body v blízkosti s približne rovnakou funkčnou hodnotou (dotyčnica je takmer rovnobežná s osou x a pretína ju vo veľmi vzdialenom bode od očakávaného koreňa). Nie je vhodné voliť ani boli v blízkosti extrému funkcie, alebo v blízkosti inflexného bodu.

Newtonova metóda

Na základe separácie koreňov vieme, že na intervale [1.5, 2.5] sa nachádza práve jeden koreň.
Potrebné je overiť splenie podmienok riešiteľnosti

- $f(x)$ je spojitá na [1.5, 2.5]
- $f(x)$ je monotónna na [1.5, 2.5], buď rastúca alebo klesajúca na celom intervale, kde úlohu riešime. V našom prípade je $f(x)$ je rastúca.
- na intervale, kde úlohu riešime sa nachádza PRÁVE JEDEN KOREŇ.

Vidíme to z grafu funkcie, alebo to môžeme overiť splnením podmienky $f(1.5) * f(2.5) < 0$

- v okolí koreňa existuje derivácia $f'(x)$ - nakreslíme graf derivácie funkcie.

Pri Newtonovej metóde je potrebné zvoliť jeden štartovací bod, iteračná schéma má tvar

$x_0 =. ....$

$$x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$$

Podmienky použitia metódy budú možno porušené v okolí bodu 2.5 - vidíme to z grafu funkcie, ktorý sme nakreslili na intervale [0.1, 10], preto je lepšie voliť štartujúci bod menší ako 2. Pre ďalší výpočet zvolíme štartujúce hodnoty $x_0 = 1$.

In[22]:=
```
Plot[f'[x], {x, 1.5, 2.5}]
```

Out[22]=


Vidíme, že na intervale [1.5, 2.5] existuje derivácia funkcie a nenadobúda hodnoty blízke nule (to by mohlo viesť ku zlej podmienenosti úlohy).

In[49]:=
```
Clear[f, x]
f[x_] = x^2 + Log[x] - 10 / x;
x[0] = 1;
pocetopakovani = 10;
tolerancia = 10^(-10);
```

In[54]:=
```
Do[x[i + 1] = x[i] - f[x[i]] / f'[x[i]] // N;
 If[Abs[x[i + 1] - x[i]] < tolerancia, Break[]],
 {i, 0, pocetopakovani}]
```

In[55]:=
```
TableForm[Table[{i, NumberForm[x[i], 10], Abs[x[i + 1] - x[i]]}, {i, 0, pocetopakovani}],
    TableHeadings → {None, {"i", "xi", "|xi+1-xi|"}},
    TableSpacing → {1, 5}]
```

Out[55]//TableForm=

i	x_i	$\|x_{i+1}-x_i\|$
0	1	0.692308
1	1.692307692	0.337352
2	2.029659343	0.0142633
3	2.043922603	9.00229×10^{-6}
4	2.043931605	3.38085×10^{-12}
5	2.043931605	Abs[-2.04393 + x[6]]
6	x[6]	Abs[-x[6] + x[7]]
7	x[7]	Abs[-x[7] + x[8]]
8	x[8]	Abs[-x[8] + x[9]]
9	x[9]	Abs[-x[9] + x[10]]
10	x[10]	Abs[-x[10] + x[11]]

Kvôli istote sa pozrime ešte na presný tvar koreňa, pomocou príkazu: `NumberForm [číslo, počet miest]`

In[57]:=
```
NumberForm[x[4], 16]
```

Out[57]//NumberForm=
2.043931605058533

Výpočet realizovaný Newtonovou metódou nám našiel približnú hodnotu koreňa rovnice x_4 = 2.053931605058533. Uvedomte si, že je to len PRIBLIŽNÁ hodnota koreňa, skutočnú hodnotu koreňa nepoznáme, vieme len, že je veľmi blízko tejto vypočítanej hodnoty.
Vypočítajme preto odhad absolútnej chyby aproximácie podľa vzorca pre výpočet odhadu chyby. Tento vzorec nám dáva informáciu o tom, ako blízko sa nachádza nami vypočítaná približná hodnota koreňa x_4 od skutočného koreňa rovnice. Použijeme vzťah z prednášky.

$$|x_i - \alpha| \le \frac{|f(x_i)|}{m}, \qquad \text{kde } m = \min\,|f'(x_i)| \text{ na intervale } [a, b]$$

Najskôr si nakreslíme deriváciu funkcie $f(x)$ na intervale, na ktorom realizujeme výpočet - [1.5, 2.5].

In[22]:=
```
Plot[f'[x], {x, 1.5, 2.5}]
```

Out[22]=


Vidíme, že na intervale [1.5, 2.5] derivácia funkcie $-\ f'(x)$ nadobúda najmenšiu hodnotu niekde v okolí bodu 2.217. Zistili sme to systémom zistím a vidím, odčítaním pomocou zamerania bodu z intervalu.

In[17]:=
```
f'[2.217] // N
```

Môžeme použiť aj funkciu programu *Mathematica* na nájdenie minima - to bude určite presnejšie, ako náš odhad. My ale hľadáme odhad, takže na presnosti až tak veľmi nezáleží a príkaz *Mathematice* môže v mnohých situáciách zlyhať.

In[23]:=
```
FindMinimum[f'[x], {x, 1.5, 2.5}]
```

Out[23]= {6.91933, {x → 2.23176}}

Vypočítajme odhad absolútnej chyby aproximácie podľa vzorca pre výpočet odhadu chyby - dosadíme do nasledovného vzťahu.

$|x_i - \alpha| \leq \frac{|f'(x_i)|}{m}$, kde $m = \min\ |\ f'(x_i)\ |$ na intervale $[a, b]$

In[34]:=
```
Abs[f[x[4]]] / f'[2.217]
```

Out[34]= 3.40575×10^{-12}

Vidíme, že nami vypočítaná aproximácia koreňa rovnice sa od skutočného koreňa líší rádo 10^{-12}, čo je viac ako bola požadovaná presnosť výpočtu a preto môžeme považovať nami vypočítanú aproximáciu koreňa za použiteľnú.

Nami vypočítaný výsledok môžeme porovnať aj s riešením nájdeným pomocou príkazu FindRoot [ ]. Je to príkaz systému *Mathematica*. Ak parametre výpočtu je potrebné uviesť jeden štartovací bod - vtedy bude *Mathematica* používať pre výpočet "vylepšenú Newtonovu" metódu .

In[38]:=
```
FindRoot[f[x], {x, 1}]
```

Out[38]= {x → 2.04393}

V zadaní ešte žiadali nájsť bod, ktorý nie je vhodnou voľbou štartovacieho bodu pre Newtonovu metódu - napr. 0.1   Pri NEWTONOVEJ METODE nie je vhodné voliť štartovací bod v blízkosti extrémov (dotyčnica je takmer rovnobežná s osou x a pretína ju vo veľmi vzdialenom bode od očakávaného koreňa), alebo v blízkosti inflexných bodov (presečníky dotyčníc oscilujú).   Nie je vhodné voliť ani boli v blízkosti extrému funkcie, alebo v blízkosti inflexného bodu.

■ Záver

NEWTONOVA METÓDA rýchlejšie konverguje ku hľadanému koreňu (na dosiahnutie aproximácie koreňa so žiadanou toleranciou boli potrebné 4 iterácie). METÓDA TETÍV ich potrebovala 6. Na druhej strane však metóda tetív spočítala koreň presnejšie - vidíme to na základe výpočtu v odhade chyby.

Programy

■ Metóda tetív

```
Clear[f, x]
f[x_] =;
x[0] =;
x[1] =;
pocetopakovani =;
tolerancia =;
Do[x[i + 1] = x[i] - f[x[i]] * (x[i] - x[i - 1]) / (f[x[i]] - f[x[i - 1]]) // N;
 If[Abs[x[i + 1] - x[i]] < tolerancia, Break[]],
 {i, 1, pocetopakovani}]

TableForm[Table[{i, NumberForm[x[i], 10], Abs[x[i + 1] - x[i]]}, {i, 0, pocetopakovani}],
     TableHeadings → {None, {"i", "xi", "|xi+1-xi|"}},
     TableSpacing → {1, 5}]
```

Newtonova metóda

```
Clear[f, x]
f[x_] =;
x[0] =;
pocetopakovani =;
tolerancia =;
Do[x[i + 1] = x[i] - f[x[i]] / f'[x[i]] // N;
 If[Abs[x[i + 1] - x[i]] < tolerancia, Break[]],
 {i, 0, pocetopakovani}]

TableForm[Table[{i, NumberForm[x[i], 10], Abs[x[i + 1] - x[i]]}, {i, 0, pocetopakovani}],
     TableHeadings → {None, {"i", "xi", "|xi+1-xi|"}},
     TableSpacing → {1, 5}]
```

Programy pre lenivých programátorov

Na písomke nie je potrebné používať plnohodnotné programy, obsahujúce perfektný výstup, zastavovaciu podmienku... Podstatné je získať správny výsledok. Preto je možné programy zjednodušiť - nasledujúca verzia jeúplne minimalistická, vyžaduje ale správnu interpretáciu vypočítaných výsledkov.

Metóda tetív

```
Clear[f, x]
f[x_] =;
x[0] =;
x[1] =;

Do[x[i + 1] = x[i] - f[x[i]] * (x[i] - x[i - 1]) / (f[x[i]] - f[x[i - 1]]) // N;
 Print["i=", i, "          x[i]=", x[i] ] ,
 {i, 1, 20}]
```

Newtonova metóda

```
Clear[f, x]
f[x_] =;
x[0] =;

Do[x[i + 1] = x[i] - f[x[i]] / f'[x[i]] // N;
 Print["i=", i, "          x[i]=", x[i]] ,
 {i, 0, 20}]
```