Príklad 1:

Je daná dátová množina. Nájdite interpolačný polynóm, ktorý najlepšie túto dátovú množinu popisuje pomocou rôznych metód uvedených na prednáške.
Nájdite interpolačný polynóm, ktorý prechádza bodmi [0,0], [1,0], [3,24], [5,120]

Interpolácia pomocou priameho výpočtu - riešenie sústavy rovníc

In[ ]:=
```
data = {{0, 0.5}, {1, 0}, {3, 24}, {5, 120}};
obr0 = ListPlot[data, PlotStyle → PointSize[0.03]]
```

Out[ ]=



Ak máme danú dátovú množinu, ktorá obsahuje 4 body, tak musíme hľadať polynóm stupňa maximálne 3, ktorý dokáže túto dátovú množinu presne popísať. Tento polynóm bude prechádzať presne zadanými 4-mi bodmi.
P(x) = $a_0 + a_1\,x + a_2\,x^2 + a_3\,x^3$
Neznáme koeficienty, ktoré musíme nájsť sú: $a_{0,}\ a_1,\ a_2,\ a_3$
Dátová množina: $[x_{1,} f_1],\ [x_2, f_2],\ [x_3, f_3],\ [x_4, f_4]$
Musíme riešiť sústavu rovníc:

$$f_1 = a_0 + a_1 . x_1 + a_2 . {x_1}^2 + a_3 . {x_1}^3$$
$$f_2 = a_0 + a_1 . x_2 + a_2 . {x_2}^2 + a_3 . {x_2}^3$$
$$f_3 = a_0 + a_1 . x_3 + a_2 . {x_3}^2 + a_3 . {x_3}^3$$
$$f_4 = a_0 + a_1 . x_4 + a_2 . {x_4}^2 + a_3 . {x_4}^3$$

Sústavu rovníc zostavíme a vyriešime

In[ ]:=
```
data={{0,0.5},{1,0},{3,24},{5,120}}
```

Out[ ]=
```
{{0, 0.5}, {1, 0}, {3, 24}, {5, 120}}
```

```
Clear[x]

In[∘]:= priamyPolynom[x_] := a + b x + c x^2 + d x^3;

In[∘]:= rovnica1 = priamyPolynom[data[[1, 1]]] == data[[1, 2]]
Out[∘]=
a == 0.5

In[∘]:= rovnica2 = priamyPolynom[data[[2, 1]]] == data[[2, 2]]
Out[∘]=
a + b + c + d == 0

In[∘]:= rovnica3 = priamyPolynom[data[[3, 1]]] == data[[3, 2]]
Out[∘]=
a + 3 b + 9 c + 27 d == 24

In[∘]:= rovnica4 = priamyPolynom[data[[4, 1]]] == data[[4, 2]]
Out[∘]=
a + 5 b + 25 c + 125 d == 120

In[∘]:= riesenie = Solve[{rovnica1, rovnica2, rovnica3, rovnica4}, {a, b, c, d}]
Out[∘]=
{{a → 0.5, b → -1.76667, c → 0.3, d → 0.966667}}

In[∘]:= vysledok[x_] = priamyPolynom[x] /. riesenie[[1]]
Out[∘]=
```

$$0.5 - 1.76667\,x + 0.3\,x^2 + 0.966667\,x^3$$

```
In[∘]:= obr1 = Plot[vysledok[x], {x, -0.5, 6}];
        Show[obr0, obr1]
Out[∘]=
```



Interpolácia pomocu Newtonovho polynómu

Nájdite Newtonow interpolačný polynóm 3. stupňa a vypočítajte jeho hodnotu v bode $x = 0.3$

```
x (i) :        0.142857    0.285714    0.428571    0.571429    0.714286    0.857143    1.

f (x (i)) :
     0.           1.38629     3.29584     5.54518     8.04719     10.7506     13.6214
----------------------------------------------------------------------------------
                                        ------
```

Riešenie :

Táto časť je určená pre učiteľa, aby vygeneroval pekné zadania na písomku

```
f[i_] = i * Log[i];
data = Table[{i / 7., f[i] // N}, {i, 1, 4}];
```

In[ ]:=
```
Table[{i / 7., f[i] // N}, {i, 1, 7}] // Transpose // TableForm
```

Out[ ]//TableForm=

0.142857	0.285714	0.428571	0.571429	0.714286	0.857143	1.
0.	1.38629	3.29584	5.54518	8.04719	10.7506	13.6214

Riešenie, ak budeme uvažovať celú zadanú dátovú množinu - študent:

študentské časť - začína definovaním množiny dát, ktorú spracovávame

In[4]:=
```
dataFull = {{0.142857, 0.}, {0.285714, 1.38629}, {0.428571, 3.29584},
  {0.571429, 5.54518}, {0.714286, 8.04719}, {0.857143, 10.7506}, {1, 13.6214}}
```

Out[4]=
```
{{0.142857, 0.}, {0.285714, 1.38629}, {0.428571, 3.29584},
 {0.571429, 5.54518}, {0.714286, 8.04719}, {0.857143, 10.7506}, {1, 13.6214}}
```

In[ ]:=
```
x = Transpose[dataFull] [[1]]
y = Transpose[dataFull] [[2]]
```

Out[ ]=
```
{0.142857, 0.285714, 0.428571, 0.571429, 0.714286, 0.857143, 1}
```

Out[ ]=
```
{0., 1.38629, 3.29584, 5.54518, 8.04719, 10.7506, 13.6214}
```

In[ ]:=
```
pocetdat = Length[x]
```

Out[ ]=
```
7
```

In[ ]:=
```
Clear[F]
```

In[∘]:=
```
F[i_, 0] := x⟦i⟧;
F[i_, 1] := y⟦i⟧;
F[i_, j_] := (F[i, j - 1] - F[i - 1, j - 1]) / (x⟦i⟧ - x⟦i - j + 1⟧)
Table[F[i, j], {i, 1, pocetdat}, {j, 0, i}] // TableForm
```

Out[∘]//TableForm=

0.142857	0.						
0.285714	1.38629	9.70404					
0.428571	3.29584	13.3669	12.8199				
0.571429	5.54518	15.7453	8.32446	-10.4893			
0.714286	8.04719	17.5141	6.19079	-4.97855	9.64389		
0.857143	10.7506	18.9239	4.93431	-2.93179	3.58183	-8.48688	
1	13.6214	20.0956	4.10106	-1.94424	1.7282	-2.59508	6.87377

In[∘]:=
```
Clear[X]
```

In[∘]:=
```
F[1, 1] +
 Apply[Plus, Table[F[j + 1, j + 1] * Product[(X - x⟦i⟧), {i, 1, j}], {j, 1, pocetdat - 1}]]
```

Out[∘]=

```
0. + 9.70404 (-0.142857 + X) + 12.8199 (-0.285714 + X) (-0.142857 + X) -
 10.4893 (-0.428571 + X) (-0.285714 + X) (-0.142857 + X) +
 9.64389 (-0.571429 + X) (-0.428571 + X) (-0.285714 + X) (-0.142857 + X) -
 8.48688 (-0.714286 + X) (-0.571429 + X) (-0.428571 + X) (-0.285714 + X) (-0.142857 + X) +
 6.87377 (-0.857143 + X) (-0.714286 + X) (-0.571429 + X)
  (-0.428571 + X) (-0.285714 + X) (-0.142857 + X)
```

In[∘]:=
```
NewtonFull[X_] = F[1, 1] + Apply[Plus,
    Table[F[j + 1, j + 1] * Product[(X - x⟦i⟧), {i, 1, j}], {j, 1, pocetdat - 1}]] // Expand
```

Out[∘]=

$-0.480483 - 1.24071 X + 38.9157 X^2 - 53.718 X^3 + 52.3793 X^4 - 29.1082 X^5 + 6.87377 X^6$

Hodnota polynómu v žiadanom bode - vypočítame dosadením

In[∘]:=
```
NewtonFull[0.3]
```

Out[∘]=

1.55788

Hodnota derivácie polynómu v žiadanom bode - vypočítame deriváciu a následne do derivácie dosadíme požadovaný bod

In[∘]:=
```
D[NewtonFull[X], X]
D[NewtonFull[X], X] /. X → 0.142857
```

Out[∘]=

$-1.24071 + 77.8315 X - 161.154 X^2 + 209.517 X^3 - 145.541 X^4 + 41.2426 X^5$

Out[∘]=

7.14188

In[∘]:=
```
dataFull
```

Out[∘]=

```
{{0.142857, 0.}, {0.285714, 1.38629}, {0.428571, 3.29584},
 {0.571429, 5.54518}, {0.714286, 8.04719}, {0.857143, 10.7506}, {1, 13.6214}}
```

In[ ]:=
```
obr0 = ListPlot[dataFull, PlotStyle → {Red, PointSize[0.03]}];
obr2 = Plot[NewtonFull[X], {X, Min[x], Max[x]}];
obr3 = Plot[NewtonFull[X], {X, Min[x] - 1, Max[x] + 1}, PlotStyle → Gray];
Show[obr3, obr2, obr0]
```

Out[ ]=



Riešenie - študent:

študentské časť - začína definovaním množiny dát, ktorú spracovávame

In[1]:=
```
data = {{0.142857, 0.}, {0.285714, 1.38629}, {0.428571, 3.29584}, {0.571429, 5.54518}}
```

Out[1]=
```
{{0.142857, 0.}, {0.285714, 1.38629}, {0.428571, 3.29584}, {0.571429, 5.54518}}
```

In[ ]:=
```
x = Transpose[data] [[1]]
y = Transpose[data] [[2]]
pocetdat = Length[x];
```

Out[ ]=
```
{0.142857, 0.285714, 0.428571, 0.571429}
```

Out[ ]=
```
{0., 1.38629, 3.29584, 5.54518}
```

In[ ]:=
```
Clear[F]
```

In[ ]:=
```
F[i_, 0] := x[[i]];
F[i_, 1] := y[[i]];
F[i_, j_] := (F[i, j - 1] - F[i - 1, j - 1]) / (x[[i]] - x[[i - j + 1]])
Table[F[i, j], {i, 1, pocetdat}, {j, 0, i}] // TableForm
```

Out[ ]//TableForm=

0.142857	0.			
0.285714	1.38629	9.70404		
0.428571	3.29584	13.3669	12.8199	
0.571429	5.54518	15.7453	8.32446	-10.4893

In[ ]:=
```
Clear[X]
```

In[∘]:=
```
F[1, 1] +
 Apply[Plus, Table[F[j + 1, j + 1] * Product[(X - x⟦i⟧), {i, 1, j}], {j, 1, pocetdat - 1}]]
```

Out[∘]=
```
0. + 9.70404 (-0.142857 + X) + 12.8199 (-0.285714 + X) (-0.142857 + X) -
 10.4893 (-0.428571 + X) (-0.285714 + X) (-0.142857 + X)
```

In[∘]:=
```
Newton[X_] = F[1, 1] + Apply[Plus,
    Table[F[j + 1, j + 1] * Product[(X - x⟦i⟧), {i, 1, j}], {j, 1, pocetdat - 1}]] // Expand
```

Out[∘]=

$-0.679543 + 1.85506\, X + 21.8108\, X^2 - 10.4893\, X^3$

Hodnota polynómu v žiadanom bode - vypočítame dosadením

In[∘]:=
```
Newton[0.3]
```

Out[∘]=
```
1.55673
```

Hodnota derivácie polynómu v žiadanom bode - vypočítame deriváciu a následne do derivácie dosadíme požadovaný bod

In[∘]:=
```
D[Newton[X], X]
D[Newton[X], X] /. X → 0.142857
```

Out[∘]=

$1.85568 + 43.6173\, X - 31.4617\, X^2$

Out[∘]=
```
7.44464
```

In[∘]:=
```
obr1 = ListPlot[data, PlotStyle → {Green, PointSize[0.04]}];
obr4 = Plot[Newton[X], {X, Min[x], Max[x]}];
obr5 = Plot[Newton[X], {X, Min[x] - 1, Max[x] + 1}, PlotStyle → Gray];
Show[obr5, obr4, obr1]
```

Out[∘]=



Ak do jedného obrázku zakreslíme oba polynómy - aj ten, ktorý sme získali analýzou úplnej dátovej množiny, aj ten, ktorý sme získali len z výberovej podmnožiny dát - vidíme, že mimo dátovej množiny sa polynómy zásadným spôsobom líšia

In[ ]:= `Show[obr5, obr3, obr1, obr0]`

Out[ ]=



In[2]:= `InterpolatingPolynomial[data, x]`

Out[2]= $5.54518 + (12.9387 + (9.82293 - 10.4893\,(-0.428571 + x))\,(-0.142857 + x))\,(-0.571429 + x)$

In[3]:= `InterpolatingPolynomial[data, x] // Expand`

Out[3]= $-0.679543 + 1.85506\,x + 21.8108\,x^2 - 10.4893\,x^3$

In[5]:= `InterpolatingPolynomial[dataFull, x]`

Out[5]= $13.6214 + (15.8916 + (6.89009 + (-6.20384 + (4.70407 + (-6.52295 + 6.87377\,(-0.428571 + x))\,(-0.857143 + x))\,(-0.285714 + x))\,(-0.571429 + x))\,(-0.142857 + x))\,(-1 + x)$

In[6]:= `InterpolatingPolynomial[dataFull, x] // Expand`

Out[6]= $-0.480483 - 1.24071\,x + 38.9157\,x^2 - 53.718\,x^3 + 52.3793\,x^4 - 29.1082\,x^5 + 6.87377\,x^6$