Zadanie príkladov

Máme danú dátovú množinu

. . . .

Aproximácia pomocou priamky $f: y = \mathrm{Ax} + B$

Metódou najmenších štvorcov nájdite priamku $y = f(x) = \mathrm{Ax} + B$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Nakreslite graf zadanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, B) = \sum_{i=1}^{n} (A\, x_i + B - y_i)^2$
- Napíšte sústavu normálových rovníc (2 rovnice)
- Nájdite riešenie sústavy normálových rovníc.
- Napíšte rovnicu priamky, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú priamku.

Aproximácia pomocou paraboly $f: y = \mathrm{Ax}^2 + \mathrm{Bx} + C$

Metódou najmenších štvorcov nájdite parabolu $y = f(x) = \mathrm{Ax}^2 + \mathrm{Bx} + C$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Nakreslite graf zadanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, B, C) = \sum_{i=1}^{n} \left(A\, x_i^2 + B\, x_i + C - y_i\right)^2$
- Napíšte sústavu normálových rovníc (3 rovnice)
- Nájdite riešenie sústavy normálových rovníc.
- Napíšte rovnicu paraboly, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú parabolu.

Aproximácia pomocou exponenciálnej funkcie - nelineárna metóda $f: y = C\, e^{A\, x}$

Metódou najmenších štvorcov nájdite exponenciálnu funkciu $f: y = C\, e^{A\, x}$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Nakreslite graf zadanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, C) = \sum_{i=1}^{n} \left(C\, e^{A\, x_i} - y_i\right)^2$
- Napíšte sústavu normálových rovníc (2 rovnice), pozor rovnice sú nelineárne
- Nájdite riešenie sústavy normálových rovníc - sústavu riešte ako nelineárnu sústavu rovníc.
- Napíšte rovnicu exponenciálnej funkcie, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú exponenciálnu funkciu.

Aproximácia pomocou exponenciálnej funkcie - metóda linearizácie dát $f: y = C\, e^{A\, x}$

Metódou najmenších štvorcov nájdite exponenciálnu funkciu $f: y = C\, e^{A\, x}$ pre dáta dané tabuľkou

$E_2(f)$

$f: y = C e^{A x}$

a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Popíšte princíp metódy linearizácie dát v tomto konkrétnom príklade.
- Urobte linearizáciu na zadanej dátovej množine
- Nakreslite graf linearizovanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, B) = \sum_{i=1}^{n} (A X_i + B - Y_i)^2$ - pre linearizované dáta
- Napíšte sústavu normálových rovníc (2 rovnice)
- Nájdite riešenie sústavy normálových rovníc a dopočítajte všetky potrebné parametre
- Napíšte rovnicu exponenciálnej funkcie, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú exponenciálnu funkciu.

Ktorá z týchto aproximácií je najlepšia? Zdôvodnite svoje rozhodnutie.

Príklad 1: Aproximácia priamkou / parabolou

Máme danú dátovú množinu

x_i	-2	-1	0	1	2
y_i	1	2	3	3	4

Aproximácia pomocou priamky y = Ax + B

Máme danú dátovú množinu

x_i	-2	-1	0	1	2
y_i	1	2	3	3	4

Aproximácia pomocou priamky $y = f(x) = Ax + B$

Metódou najmenších štvorcov nájdite priamku $y = f(x) = Ax + B$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Nakreslite graf zadanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, B) = \sum_{i=1}^{n} (A x_i + B - y_i)^2$
- Napíšte sústavu normálových rovníc (2 rovnice)
- Nájdite riešenie sústavy normálových rovníc.
- Napíšte rovnicu priamky, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú priamku.

```
In[1]:= Clear[data]
data = {{-2, 1}, {-1, 2}, {0, 3}, {1, 3}, {2, 4}};
```

```
In[3]:= n = Length[data];
```

Z dátovej množiny vyberieme x-ové súradnice a y-ové súradnice jednotlivých bodov

In[4]:=
```
x = Transpose[data][[1]];
y = Transpose[data][[2]];
```

Nakreslíme zadané dáta

In[6]:=
```
obr1 = ListPlot[data, PlotStyle → PointSize[0.02]]
```

Out[6]=


Definujeme funkciu chyby pre priamku $E(A,\ B) = \sum_{i=1}^{n}(p_n(x_i) - y_i)^2 = \sum_{i=1}^{n}(A\,x_i + B - y_i)^2$

```
Error[A_, B_] = Sum[(A * x[[i]] + B - y[[i]])^2, {i, 1, n}] // Expand
```

$39 - 14\,A + 10\,A^2 - 26\,B + 5\,B^2$

Hľadáme minimálnu hodnotu tejto funkcie - budeme riešiť sústavu normálových rovníc

$$\frac{\partial E(A,B)}{\partial A} = 0$$
$$\frac{\partial E(A,B)}{\partial B} = 0$$

Na základe prednášok vieme, že v tomto bode bude funkcia chyby nadobúdať nielen svoj extrém, ale že to bude minimum.

```
D[Error[A, B], A] == 0
```

$-14 + 20\,A == 0$

```
D[Error[A, B], B] == 0
```

$-26 + 10\,B == 0$

Sústavu rovníc budeme riešiť. Je to lineárna sústava rovníc, môžeme použiť príkaz `Solve[]`

```
riesenie = Solve[{D[Error[A, B], A] == 0, D[Error[A, B], B] == 0}, {A, B}]
```

$$\left\{\left\{A \to \frac{7}{10},\ B \to \frac{13}{5}\right\}\right\}$$

Vyčistíme premennú x a definujeme hľadanú priamku. Ako parametre priamky dosadíme vypočítané hodnoty A a B.

```
Clear[x]
priamka[x_] = A * x + B /. riesenie[[1]]
```

$$\frac{13}{5} + \frac{7\,x}{10}$$

Priamku nakreslíme.

```
obr2 = Plot[priamka[x], {x, -3, 3}]
```



Výsledok vizualizujeme spolu so zadanou dátovou množinou

```
Show[obr1, obr2]
```



Veľkosť chyby, ktorej sme sa dopustili pri výpočte sa dá vypočítať veľmi jednoducho - je to len hodnota funkcie chyby pre zadané parametre *A* a *B*. Parametre môžeme dosadiť ako výsledok riešenia, alebo priamo dosadíme konkrétne vypočítané hodnoty.

```
Error[A, B] /. riesenie⟦1⟧ // N
```

```
0.3
```

```
Error[7 / 10, 13 / 5] // N
```

```
0.3
```

Aproximácia pomocou paraboly $y = Ax^2 + Bx + C$

Máme danú dátovú množinu

x_i	-2	-1	0	1	2
y_i	1	2	3	3	4

Aproximácia pomocou paraboly $y = f(x) = Ax^2 + Bx + C$

Metódou najmenších štvorcov nájdite parabolu $y = f(x) = Ax^2 + Bx + C$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Nakreslite graf zadanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, B, C) = \sum_{i=1}^{n} \left(A x_i^2 + B x_i + C - y_i\right)^2$
- Napíšte sústavu normálových rovníc (3 rovnice)
- Nájdite riešenie sústavy normálových rovníc.
- Napíšte rovnicu paraboly, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$

- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú parabolu.

Ktorá z týchto aproximácií je lepšia? Zdôvodnite.

```
Clear[data, x, y]
data = {{-2, 1}, {-1, 2}, {0, 3}, {1, 3}, {2, 4}};
```

```
n = Length[data];
```

Z dátovej množiny vyberieme x-ové súradnice a y-ové súradnice jednotlivých bodov

```
x = Transpose[data][[1]];
y = Transpose[data][[2]];
```

Nakreslíme zadané dáta

```
obr1 = ListPlot[data, PlotStyle → PointSize[0.02]]
```



Definujeme funkciu chyby pre parabolu $E(A, B, C) = \sum_{i=1}^{n} (p_n(x_i) - y_i)^2 = \sum_{i=1}^{n} \left(A x_i^2 + B x_i + C - y_i\right)^2$

Hľadáme minimálnu hodnotu tejto funkcie - budeme riešiť sústavu TROCH normálových rovníc

$$\frac{\partial E(A,B,C)}{\partial A} = 0$$
$$\frac{\partial E(A,B,C)}{\partial B} = 0$$
$$\frac{\partial E(A,B,C)}{\partial C} = 0$$

Na základe prednášok vieme, že v tomto bode bude funkcia chyby nadobúdať nielen svoj extrém, ale že to bude minimum.

```
Clear[Error, A, B]
Error[A_, B_, C_] = Sum[(A * x[[i]]^2 + B * x[[i]] + C - y[[i]])^2, {i, 1, n}] // Expand
```

$39 - 50\,A + 34\,A^2 - 14\,B + 10\,B^2 - 26\,C + 20\,A\,C + 5\,C^2$

```
D[Error[A, B, C], A] == 0
```

$-50 + 68\,A + 20\,C == 0$

```
D[Error[A, B, C], B] == 0
```

$-14 + 20\,B == 0$

```
D[Error[A, B, C], C] == 0
```

$-26 + 20\,A + 10\,C == 0$

Sústavu rovníc budeme riešiť. Je to lineárna sústava rovníc, môžeme použiť príkaz `Solve[]`

```
riesenie = Solve[{D[Error[A, B, C], A] == 0,
   D[Error[A, B, C], B] == 0, D[Error[A, B, C], C] == 0}, {A, B, C}]
```

$$\left\{\left\{A \to -\frac{1}{14}, B \to \frac{7}{10}, C \to \frac{96}{35}\right\}\right\}$$

Vyčistíme premennú x a definujeme hľadanú parabolu. Ako parametre paraboly dosadíme vypočítané hodnoty A, B a C.

```
Clear[x]
parabola[x_] = A * x^2 + B * x + C /. riesenie[[1]]
```

$$\frac{96}{35} + \frac{7\,x}{10} - \frac{x^2}{14}$$

Parabolu nakreslíme.

```
obr2 = Plot[parabola[x], {x, -3, 3}]
```



Výsledok vizualizujeme spolu so zadanou dátovou množinou

```
Show[obr1, obr2]
```



Veľkosť chyby, ktorej sme sa dopustili pri výpočte sa dá vypočítať veľmi jednoducho - je to len hodnota funkcie chyby pre zadané parametre A, B a C. Parametre môžeme dosadiť ako výsledok riešenia, alebo priamo dosadíme konkrétne vypočítané hodnoty.

```
Error[A, B, C] /. riesenie[[1]] // N
```

0.228571

```
Error[-1 / 14, 7 / 10, 96 / 35] // N
```

0.228571

Záver

Na základe porovnania veľkosti chýb vidíme, že aproximácia zadanej dátovej množiny pomocou paraboly je lepšia - dopustili sme sa menšej chyby.

Príklad 2: Aproximácia exponenciálnou funkciou $y = C\,e^{A x}$ a parabolou $y = Ax^2 + Bx + C$

Príklad 2: Aproximácia exponenciálnou funkciou $y = C e^{A x}$ a parabolou $y = \mathrm{A}x^2 + \mathrm{B}x + C$

Máme danú dátovú množinu

x_i	1	2	3	4	5
y_i	0.6	1.9	4.3	7.6	12.6

Aproximácia pomocou paraboly $f: y = \mathrm{A}x^2 + \mathrm{B}x + C$

Metódou najmenších štvorcov nájdite parabolu $y = f(x) = \mathrm{A}x^2 + \mathrm{B}x + C$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Nakreslite graf zadanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, B, C) = \sum_{i=1}^{n} \left(A x_i^2 + B x_i + C - y_i\right)^2$
- Napíšte sústavu normálových rovníc (3 rovnice)
- Nájdite riešenie sústavy normálových rovníc.
- Napíšte rovnicu paraboly, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú parabolu.

Aproximácia pomocou exponenciálnej funkcie - nelineárna metóda $f: y = C e^{A x}$

Metódou najmenších štvorcov nájdite exponenciálnu funkciu $f: y = C e^{A x}$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Nakreslite graf zadanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, C) = \sum_{i=1}^{n} \left(C e^{A x_i} - y_i\right)^2$
- Napíšte sústavu normálových rovníc (2 rovnice), pozor rovnice sú nelineárne
- Nájdite riešenie sústavy normálových rovníc - sústavu riešte ako nelineárnu sústavu rovníc.
- Napíšte rovnicu exponenciálnej funkcie, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú exponenciálnu funkciu.

Aproximácia pomocou exponenciálnej funkcie - metóda linearizácie dát $f: y = C e^{A x}$

Metódou najmenších štvorcov nájdite exponenciálnu funkciu $f: y = C e^{A x}$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Popíšte princíp metódy linearizácie dát v tomto konkrétnom príklade.
- Urobte linearizáciu na zadanej dátovej množine
- Nakreslite graf linearizovanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, B) = \sum_{i=1}^{n} (A X_i + B - Y_i)^2$ - pre linearizované dáta
- Napíšte sústavu normálových rovníc (2 rovnice)
- Nájdite riešenie sústavy normálových rovníc a dopočítajte všetky potrebné parametre
- Napíšte rovnicu exponenciálnej funkcie, ktorá najlepšie aproximuje zadanú dátovú množinu.

$E_2(f)$

$E(A, B) = \sum_{i=1}^{n} (A X_i + B - Y_i)^2$

- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú exponenciálnu funkciu.

Ktorá z týchto aproximácií je najlepšia? Zdôvodnite svoje rozhodnutie.

Aproximácia pomocou paraboly $y = Ax^2 + Bx + C$

Máme danú dátovú množinu

x_i	1	2	3	4	5
y_i	0.6	1.9	4.3	7.6	12.6

Aproximácia pomocou paraboly $f: y = Ax^2 + Bx + C$

Metódou najmenších štvorcov nájdite parabolu $y = f(x) = Ax^2 + Bx + C$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Nakreslite graf zadanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, B, C) = \sum_{i=1}^{n} \left(A x_i^2 + B x_i + C - y_i\right)^2$
- Napíšte sústavu normálových rovníc (3 rovnice)
- Nájdite riešenie sústavy normálových rovníc.
- Napíšte rovnicu paraboly, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú parabolu.

```
Clear[data]
data = {{1, 0.6}, {2, 1.9}, {3, 4.3}, {4, 7.6}, {5, 12.6}};
```

```
n = Length[data];
```

Z dátovej množiny vyberieme x-ové súradnice a y-ové súradnice jednotlivých bodov

```
x = Transpose[data][[1]];
y = Transpose[data][[2]];
```

Nakreslíme zadané dáta

```
obr1 = ListPlot[data, PlotStyle → PointSize[0.02]]
```



Definujeme funkciu chyby pre parabolu $E(A, B, C) = \sum_{i=1}^{n} (p_n(x_i) - y_i)^2 = \sum_{i=1}^{n} \left(A x_i^2 + B x_i + C - y_i\right)^2$

Hľadáme minimálnu hodnotu tejto funkcie - budeme riešiť sústavu TROCH normálových rovníc

$$\frac{\partial E(A,B,C)}{\partial A} = 0$$
$$\frac{\partial E(A,B,C)}{\partial B} = 0$$
$$\frac{\partial E(A,B,C)}{\partial C} = 0$$

Na základe prednášok vieme, že v tomto bode bude funkcia chyby nadobúdať nielen svoj extrém, ale že to bude minimum.

```
Clear[Error, A, B]
Error[A_, B_, C_] = Sum[(A * x[[i]]^2 + B * x[[i]] + C - y[[i]])^2, {i, 1, n}] // Expand
```

$238.98 - 967.\,A + 979\,A^2 - 221.4\,B + 450\,A\,B + 55\,B^2 - 54.\,C + 110\,A\,C + 30\,B\,C + 5\,C^2$

```
D[Error[A, B, C], A] == 0
```

$-967. + 1958\,A + 450\,B + 110\,C == 0$

```
D[Error[A, B, C], B] == 0
```

$-221.4 + 450\,A + 110\,B + 30\,C == 0$

```
D[Error[A, B, C], C] == 0
```

$-54. + 110\,A + 30\,B + 10\,C == 0$

Sústavu rovníc budeme riešiť. Je to lineárna sústava rovníc, môžeme použiť príkaz `Solve[]`

```
riesenie = Solve[{D[Error[A, B, C], A] == 0,
   D[Error[A, B, C], B] == 0, D[Error[A, B, C], C] == 0}, {A, B, C}]
```

$\{\{A \to 0.592857,\ B \to -0.587143,\ C \to 0.64\}\}$

Vyčistíme premennú x a definujeme hľadanú parabolu. Ako parametre paraboly dosadíme vypočítané hodnoty A, B a C.

```
Clear[x]
parabola[x_] = A * x^2 + B * x + C /. riesenie[[1]]
```

$0.64 - 0.587143\,x + 0.592857\,x^2$

Parabolu nakreslíme.

```
obr2 = Plot[parabola[x], {x, -1, 6}]
```



Výsledok vizualizujeme spolu so zadanou dátovou množinou

```
Show[obr1, obr2]
```



Veľkosť chyby, ktorej sme sa dopustili pri výpočte sa dá vypočítať veľmi jednoducho - je to len hodnota funkcie chyby pre zadané parametre A, B a C. Parametre môžeme dosadiť ako výsledok riešenia, alebo priamo dosadíme konkrétne vypočítané hodnoty.

Veľkosť chyby, ktorej sme sa dopustili pri výpočte sa dá vypočítať veľmi jednoducho - je to len hodnota funkcie chyby pre zadané parametre A, B a C. Parametre môžeme dosadiť ako výsledok riešenia, alebo priamo dosadíme konkrétne vypočítané hodnoty.

```
Error[A, B, C] /. riesenie[[1]] // N
```

```
0.0502857
```

Nelineárna aproximácia pomocou exponenciálnej funkcie $y = C e^{A x}$

Máme danú dátovú množinu

x_i	1	2	3	4	5
y_i	0.6	1.9	4.3	7.6	12.6

Aproximácia pomocou exponenciálnej funkcie - nelineárna metóda $f: y = C e^{A x}$
Metódou najmenších štvorcov nájdite exponenciálnu funkciu $f: y = C e^{A x}$ pre dáta dané tabuľkou a vypočítajte akej chyby $E_2(f)$ ste sa dopustili.

- Definujte zadanú dátovú množinu.
- Nakreslite graf zadanej dátovej množiny
- Napíšte konkrétny tvar funkcie chyby $E(A, C) = \sum_{i=1}^{n} \left(C e^{A x_i} - y_i\right)^2$
- Napíšte sústavu normálových rovníc (2 rovnice), pozor rovnice sú nelineárne
- Nájdite riešenie sústavy normálových rovníc - sústavu riešte ako nelineárnu sústavu rovníc.
- Napíšte rovnicu exponenciálnej funkcie, ktorá najlepšie aproximuje zadanú dátovú množinu.
- Vypočítajte veľkosť chyby aproxmácie $E_2(f)$
- Urobte vizualizáciu prolému, do jedného obrázka nakreslite zadanú dátovú množinu, aj vami nájdenú exponenciálnu funkciu.

```
Clear[data]
data = {{1, 0.6}, {2, 1.9}, {3, 4.3}, {4, 7.6}, {5, 12.6}};
```

```
n = Length[data];
```

Z dátovej množiny vyberieme x-ové súradnice a y-ové súradnice jednotlivých bodov

```
x = Transpose[data][[1]];
y = Transpose[data][[2]];
```

Nakreslíme zadané dáta

```
obr1 = ListPlot[data, PlotStyle → PointSize[0.02]]
```



Definujeme funkciu chyby pre exponenciálnu funkciu $E(A, C) = \sum_{i=1}^{n} (f(x_i) - y_i)^2 = \sum_{i=1}^{n} \left(C\ e^{A\ x_i} - y_i\right)^2$
Všimnite si, že táto funkcia je nelineárna

Hľadáme minimálnu hodnotu tejto funkcie - budeme riešiť sústavu dvoch NELINEÁRNYCH normálových rovníc

$$\frac{\partial E(A,C)}{\partial A} = 0$$
$$\frac{\partial E(A,C)}{\partial C} = 0$$

Hľadáme minimálnu hodnotu tejto funkcie - budeme riešiť sústavu dvoch NELINEÁRNYCH normálových rovníc

$$\frac{\partial E(A,C)}{\partial A} = 0$$
$$\frac{\partial E(A,C)}{\partial C} = 0$$

```
Error[A_, C_] = Sum[(C * Exp[ A * x[[i]]] - y[[i]])^2, {i, 1, n}] // Expand
```

$238.98 - 1.2\,C\,e^{A} - 3.8\,C\,e^{2A} + C^2\,e^{2A} - 8.6\,C\,e^{3A} - 15.2\,C\,e^{4A} + C^2\,e^{4A} - 25.2\,C\,e^{5A} + C^2\,e^{6A} + C^2\,e^{8A} + C^2\,e^{10A}$

```
D[Error[A, C], A] == 0 // Simplify
```

$C\,e^{A}\left(-1.2 + (-7.6 + 2C)\,e^{A} - 25.8\,e^{2A} + (-60.8 + 4C)\,e^{3A} - 126.\,e^{4A} + 6C\,e^{5A} + 8C\,e^{7A} + 10C\,e^{9A}\right) == 0$

```
D[Error[A, C], C] == 0 // Simplify
```

$e^{A}\left(-1.2 + (-3.8 + 2C)\,e^{A} - 8.6\,e^{2A} + (-15.2 + 2C)\,e^{3A} - 25.2\,e^{4A} + 2C\,e^{5A} + 2C\,e^{7A} + 2C\,e^{9A}\right) == 0$

Najskôr sa pokúsime vyriešiť tento systém pomocou príkazu `Solve[]`. Tento systém však nie je možné riešiť priamo - *Mathematica* nás upozorňuje, že nenašla všetky riešenia a vidíme, že mnohé riešenia sú komplexné. Medzi riešeniami je možné v tomto prípade nájsť jedno reálne riešenie.

```
riesenie = Solve[{D[Error[A, C], A] == 0, D[Error[A, C], C] == 0}, {A, C}]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

```
{{C → -4.60134 - 2.20101 i, A → 0.239477+ 1.70658 i, e^A → -0.171991 + 1.25889 i},
 {C → -4.60134 + 2.20101 i, A → 0.239477- 1.70658 i, e^A → -0.171991 - 1.25889 i},
 {C → -0.904199, A → -1.15954 + 3.14159 i, e^A → -0.31363},
 {C → 0., A → -0.810014 - 2.54381 i, e^A → -0.367709 - 0.250367 i},
 {C → 0., A → -0.810014 + 2.54381 i, e^A → -0.367709 + 0.250367 i},
 {C → 0., A → -0.712247 - 1.43559 i, e^A → 0.0661213- 0.486064 i},
 {C → 0., A → -0.712247 + 1.43559 i, e^A → 0.0661213+ 0.486064 i},
 {C → 0.698763, A → 0.581815, e^A → 1.78928},
 {C → 0.843533- 1.08162 i, A → -1.04465 + 1.76001 i, e^A → -0.0661726 + 0.345537 i},
 {C → 0.843533+ 1.08162 i, A → -1.04465 - 1.76001 i, e^A → -0.0661726 - 0.345537 i},
 {C → 2.55242- 7.04508 i, A → 0.196007- 1.01053 i, e^A → 0.646481- 1.03054 i},
 {C → 2.55242+ 7.04508 i, A → 0.196007+ 1.01053 i, e^A → 0.646481+ 1.03054 i},
 {C → 3.84495- 0.0682088 i, A → 0.20488+ 2.37375 i, e^A → -0.882986 + 0.85252 i},
 {C → 3.84495+ 0.0682088 i, A → 0.20488- 2.37375 i, e^A → -0.882986 - 0.85252 i}}
```

Bohužiaľ vo všeobecnosti sa nám takýto systém vôbec nemusí podariť vyriešiť.

Ak príkaz `Solve[]` zlyhá, môžeme použiť na výpočet aj numerické metódy na riešenie nelineárneho systému rovníc napríklad Newtonovu metódu (nepreberali sme ju pre prípad systému rovníc, budeme sa ňou zaoberať až na inžinierskom štúdiu). Táto metóda je implementovaná do *Mathematice* pomocou príkazu `FindRoot[]`, je potrebné zvoliť dva štartovacie body. Bohužiaľ overenie podmienok použiteľnosti tejto metódy je dosť komplikované, voľba správnych štartovacích bodov je v tomto štádiu len vec náhody.

```
riesenie = FindRoot[{D[Error[A, C], A] == 0, D[Error[A, C], C] == 0}, {{A, 1}, {C, 1}}]
{A → 0.581815, C → 0.698763}
```

Vyčistíme premennú x a definujeme hľadanú exponenciálnu funkciu. Ako parametre exponenciálnej funkcie dosadíme vypočítané hodnoty A a C.

```
Clear[x]
exponential[x_] = C Exp[A x] /. riesenie
```

$0.698763\, e^{0.581815\, x}$

Exponenciálnu funkciu nakreslíme.

```
obr3 = Plot[exponential[x], {x, 0, 5}, PlotStyle → Red]
```



Výsledok vizualizujeme spolu so zadanou dátovou množinou - nakreslíme do jedného obrázka aj nájdenú parabolu, aj vypočítanú exponenciálnu funkciu.

```
Show[obr1, obr2, obr3]
```



Veľkosť chyby, ktorej sme sa dopustili pri výpočte sa dá vypočítať veľmi jednoducho - je to len hodnota funkcie chyby pre zadané parametre A a C. Parametre môžeme dosadiť ako výsledok riešenia, alebo priamo dosadíme konkrétne vypočítané hodnoty.
Poznámka - všimnite, že príkaz `FindRoot[]` nám dal riešenie len v tvare jednorozmerného poľa, preto dosadenie je možné pomocou nasledujúceho príkazu. (Nemusíme manipulovať s dvojrozmerným poľom, ako v predchádzajúcich príkladoch.)

```
Error[A, C] /. riesenie
0.862808
```

Aproximácia pomocou exponenciálnej funkcie - metóda linealizácie $y = C\, e^{A x}$

```
Clear[data]
data = {{1, 0.6}, {2, 1.9}, {3, 4.3}, {4, 7.6}, {5, 12.6}};
```

```
n = Length[data]
```

5

Z dátovej množiny vyberieme x-ové súradnice a y-ové súradnice jednotlivých bodov

```
x = Transpose[data][[1]];
y = Transpose[data][[2]];
```

Nakreslíme zadané dáta

```
obr1 = ListPlot[data, PlotStyle → PointSize[0.02]]
```



Teraz zrealizujeme linearizáciu dát. Použijeme nasledovnú myšlienku

$y = C\, e^{A x_i}$

$\ln y = \ln C\, e^{A x_i}$

$\ln y = \ln C + \ln e^{A x_i}$

$\ln y = \ln C + A\, x_i \ln e$

$\ln y = \ln C + A\, x_i$

Posledný vzťah je už lineárny, tak môžeme použiť jednoduché transformačné pravidlá

$\ln y = \ln C + A\, x_i$

$Y_i = B + A\, X_i$

Zvolená transformácia bude mať tvar

nová dátová množina	stará dátová množina
$X_i = x_i$	x_i
$Y_i = \ln y_i$	y_i

Transformačný vzťah

$Y = A\, X + B$	$y = C\, e^{A x}$

Transformácia parametrov

A - sa nemení	A
$B = \ln C$	$C = e^B$

Teraz tento teoretický postup zrealizujeme.

Nová dátová množina X bude vytvorená zo starej dátovej množiny x.

```
X = x
```

```
{1, 2, 3, 4, 5}
```

Nová dátová množina Y bude vytvorená zo starej dátovej množiny y pomocou predpisu $Y_i = \ln y_i$.

```
Y = Log[y]
```

```
{-0.510826, 0.641854, 1.45862, 2.02815, 2.5337}
```

PRE NOVÚ DÁTOVÚ MNOŽINU definujeme funkciu chyby. Po linearizácii je charakter novej dátovej množiny už lineárny, popíšeme ju priamkou.Funkcia chyby má tvar
$E(A,\ B) = \sum_{i=1}^{n}(p_n(x_i) - y_i)^2 = \sum_{i=1}^{n}(A\,x_i + B - y_i)^2$

```
Error[A_, B_] = Sum[(A * X[[i]] + B - Y[[i]])^2, {i, 1, n}] // Expand
```

$13.3335 - 51.8596\,A + 55\,A^2 - 12.303\,B + 30\,A\,B + 5\,B^2$

Hľadáme minimálnu hodnotu tejto funkcie - budeme riešiť sústavu normálových rovníc

$$\frac{\partial E(A,B)}{\partial A} = 0$$

$$\frac{\partial E(A,B)}{\partial B} = 0$$

Na základe prednášok vieme, že v tomto bode bude funkcia chyby nadobúdať nielen svoj extrém, ale že to bude minimum.

```
D[Error[A, B], A] == 0
```

$-51.8596 + 110\,A + 30\,B == 0$

```
D[Error[A, B], B] == 0
```

$-12.303 + 30\,A + 10\,B == 0$

Sústavu rovníc budeme riešiť. Je to lineárna sústava rovníc, môžeme použiť príkaz `Solve[]`

```
riesenie = Solve[{D[Error[A, B], A] == 0, D[Error[A, B], B] == 0}, {A, B}]
```

```
{{A → 0.747534, B → -1.0123}}
```

Vyčistíme premennú *x* a definujeme hľadanú exponenciálnu funkciu. Ako parametre exponenciálnej funkcie dosadíme vypočítané hodnoty *A* a *C* . Nezabudnite, že parameter *C* musíme najskôr dopočítať na základe transformačného vzťahu pomocou známeho parametra *B*.

```
Clear[x, a, b]
```

```
exponential[x_] = c Exp[a x] /. {a → 0.747534, c → Exp[-1.0123]}
```

$0.363382\,e^{0.747534\,x}$

Exponenciálnu funkciu nakreslíme.

```
obr4 = Plot[exponential[x], {x, 0, 5}]
```



Výsledok vizualizujeme spolu so zadanou dátovou množinou

```
Show[obr1, obr4]
```



Výsledok vizualizujeme spolu so zadanou dátovou množinou - nakreslíme do jedného obrázka aj nájdenú parabolu, aj vypočítanú exponenciálnu funkciu nelineárnou metódou aj exponenciálnu funkciu vypočítanú metódou linearizácie.

```
Show[obr1, obr2, obr3, obr4]
```



Veľkosť chyby, ktorej sme sa dopustili pri výpočte sa dá vypočítať veľmi jednoducho - je to len hodnota funkcie chyby pre zadané parametre *A*, *B* a C. Parametre môžeme dosadiť ako výsledok riešenia, alebo priamo dosadíme konkrétne vypočítané hodnoty.
V tomto prípade nás nezaujíma chyba, ktorej sme sa dopustili po transformácii dát, ale chyba ktorej sme sa dopustili pri aproximácii pôvodnej dátovej množiny pomocou exponenciálnej funkcie. Chybu musíme preto vypočítať presne zo základnej definície.

$$E(A,\, C) = \sum_{i=1}^{n} \left(C\, e^{A\, x_i} - y_i\right)^2$$

```
Clear[x, y]
x = Transpose[data]〚1〛;
y = Transpose[data]〚2〛;
```

```
Sum[(c Exp [a x〚i〛] - y〚i〛)^2 /. {a → 0.747534, c → Exp[-1.0123]}, {i, 1, n}]
```

```
8.10216
```

Na základe predchádzajúcich výpočtov vidíme, že

chyba pri aproximácii pomocou paraboly bola 0.0502857

chyba pri aproximácii pomocou nelineárnej exponenciálnej funkcie bola 0.862808

chyba pri aproximácii pomocou exponenciálnej funkcie metódou linearizácie bola 1.29935

Vidíme, že dáta majú pravdepodobne polynomiálny charakter, najlepšia je aproximácia pomocou paraboly.

Aproximácia pomocou predefinovaných príkazov

systému *Mathematica*

Aproximácia pomocou polynómov

Systém *Mathematica* používa predefinovaný príkaz `Fit[]`. Jeho základný tvar je

```
Fit[data,{popis aproximačnej funkcie}, premenná]
```

Aproximáciu pomocou priamky zrealizujeme príkazom

```
Fit[data,{1,x}, x]
```

Aproximáciu pomocou paraboly zrealizujeme príkazom

```
Fit[data,{1,x,x^2}, x]
```

Aproximáciu pomocou polynómu tretieho stupňa zrealizujeme príkazom

```
Fit[data,{1,x,x^2,x^3}, x]
```

Nelineárne aproximácie

Systém *Mathematica* pre nelineárne aproximácie používa predefinovaný príkaz `FindFit[]`. Jeho základný tvar je

```
FindFit[data, model, {parametre}, premenná]
```

Aproximáciu pomocou exponenciálnej funkcie zrealizujeme príkazom

```
Fit[data,c Exp[a x],{c,a]}, x]
```

Aproximácia metódou linearizácie

Mathematica nepoužíva žiadne špeciálne príkazy pre prípady transformácií dátovej množiny. Transformáciu treba uskutočniť samostatne, na základe charakteru množiny a následne môžeme použiť niektorý z predefinovaných príkazov - podľa charakteru novej množiny dát.

```
Clear[x]
data
```

{{1, 0.6}, {2, 1.9}, {3, 4.3}, {4, 7.6}, {5, 12.6}}

```
Fit[data, {1, x}, x]
```

$-3.51 + 2.97\,x$

```
Fit[data, {1, x, x^2}, x]
```

$0.64 - 0.587143\,x + 0.592857\,x^2$

```
Fit[data, {1, x, x^2, x^3}, x]
```

$-0.2 + 0.592857\,x + 0.142857\,x^2 + 0.05\,x^3$

```
FindFit[data, c Exp[a x], {a, c}, x]
```

$\{a \to 0.581815,\ c \to 0.698763\}$

Vidíme, že výsledky získané pomocou systému *Mathematica* sa zhodujú s výsledkami získanými simuláciou princípov výpočtu.